```
> restart;
> # Compute Truncation Error for a Method
> m:=4;
```

$$m := 4$$

```
> T:=y(t+h)-y(t)-h*f(t+h/2,(1/2)*(y(t+h)+y(t)));
```

$$T := y(t+h) - y(t) - h\, f\left(t + \frac{1}{2}h, \frac{1}{2}y(t+h) + \frac{1}{2}y(t)\right)$$

```
> S:=series(T,h,m);
```

$$S := (-f(t, y(t)) + \mathrm{D}(y)(t))\, h + \left(-\frac{1}{2}\mathrm{D}_1(f)(t, y(t)) - \frac{1}{2}\mathrm{D}_2(f)(t, y(t))\, \mathrm{D}(y)(t) + \frac{1}{2}\mathrm{D}^{(2)}(y)(t)\right) h^2 +$$
$$\left(-\frac{1}{8}\mathrm{D}_{1,1}(f)(t, y(t)) - \frac{1}{4}\mathrm{D}_{1,2}(f)(t, y(t))\, \mathrm{D}(y)(t) - \frac{1}{8}\mathrm{D}_{2,2}(f)(t, y(t))\, \mathrm{D}(y)(t)^2\right.$$
$$\left. - \frac{1}{4}\mathrm{D}_2(f)(t, y(t))\, \mathrm{D}^{(2)}(y)(t) + \frac{1}{6}\mathrm{D}^{(3)}(y)(t)\right) h^3 + O(h^4)$$

```
> eq[1]:=D(t->y(t))(t)=f(t,y(t));
```

$$eq_1 := \mathrm{D}(y)(t) = f(t, y(t))$$

```
> for j from 1 to m-2
  do
      eq[j+1]:=simplify(subs(seq(eq[i],i=1..j),D(unapply(eq[j],t))(t)));
  od;
```

$$eq_2 := \mathrm{D}^{(2)}(y)(t) = \mathrm{D}_1(f)(t, y(t)) + \mathrm{D}_2(f)(t, y(t))\, f(t, y(t))$$

$$eq_3 := \mathrm{D}^{(3)}(y)(t) = \mathrm{D}_{1,1}(f)(t, y(t)) + 2\,\mathrm{D}_{1,2}(f)(t, y(t))\, f(t, y(t)) + \mathrm{D}_{2,2}(f)(t, y(t))\, f(t, y(t))^2$$
$$+ \mathrm{D}_2(f)(t, y(t))\, \mathrm{D}_1(f)(t, y(t)) + \mathrm{D}_2(f)(t, y(t))^2\, f(t, y(t))$$

```
> simplify(subs(seq(eq[i],i=1..m-1),S));
```

$$\left(\frac{1}{24}\mathrm{D}_{1,1}(f)(t, y(t)) + \frac{1}{12}\mathrm{D}_{1,2}(f)(t, y(t))\, f(t, y(t)) + \frac{1}{24}\mathrm{D}_{2,2}(f)(t, y(t))\, f(t, y(t))^2\right.$$
$$\left. - \frac{1}{12}\mathrm{D}_2(f)(t, y(t))\, \mathrm{D}_1(f)(t, y(t)) - \frac{1}{12}\mathrm{D}_2(f)(t, y(t))^2\, f(t, y(t))\right) h^3 + O(h^4)$$